
Implementation of the 
LINK DECO RIGI directive 
under MPI in EUROPLEXUS 

Casadei, F. 

Larcher, M. 

Valsamos, G. 

Faucher, V. 

2018 

EUR 29206 EN 



 

 

This publication is a Technical report by the Joint Research Centre (JRC), the European Commission’s science 

and knowledge service. It aims to provide evidence-based scientific support to the European policymaking 

process. The scientific output expressed does not imply a policy position of the European Commission. Neither 

the European Commission nor any person acting on behalf of the Commission is responsible for the use that 

might be made of this publication. 

 

Contact information  

Name: Martin Larcher 

Address: JRC Ispra 

Email: martin.larcher@ec.europa.eu 

Tel.: +39-0332-789563 

 

JRC Science Hub 

https://ec.europa.eu/jrc 

 

 

JRC111615 

 

EUR 29206 EN 

 

 

 

PDF ISBN 978-92-79-82017-5 ISSN 1831-9424 doi:10.2760/80373 

 

 

Luxembourg: Publications Office of the European Union, 2018  

© European Union, 2018 

 

Reuse is authorised provided the source is acknowledged. The reuse policy of European Commission documents 
is regulated by Decision 2011/833/EU (OJ L 330, 14.12.2011, p. 39). 

For any use or reproduction of photos or other material that is not under the EU copyright, permission must be 

sought directly from the copyright holders. 

 

How to cite this report: Author(s), Title, EUR (where available), Publisher, Publisher City, Year of Publication, 

ISBN (where available), doi (where available), PUBSY No. 

 

All images © European Union 2018 

 

 

 



Implementation of the LINK DECO RIGI directive under MPI

in EUROPLEXUS

F. Casadei1, M. Larcher2, G. Valsamos2, and V. Faucher3

1Retired from JRC ELSA
2European Laboratory for Structural Assessment (ELSA), Institute for the Protection and Security of the Citizen (IPSC)

Joint Research Centre (JRC), 21027 Ispra, Italy

1 Introduction

This report presents the implementation of the LINK DECO RIGI directive under domain decomposition
(MPI version) in the EUROPLEXUS code.

EUROPLEXUS [1] (also abbreviated as EPX) is a computer code jointly developed by the French
Commissariat à l’Energie Atomique (CEA DMT Saclay) and by EC-JRC. The code application domain
is the numerical simulation of fast transient phenomena such as explosions, crashes and impacts in
complex three-dimensional fluid-structure systems. The Cast3m [2] software from CEA is used as a
pre-processor to EPX when it is necessary to generate complex meshes.

2 The RIGI link

The link of type RIGI imposes that all displacements of a set of nodes be equal, i.e. that the set of
nodes displaces rigidly in space. The name is slightly misleading since this is not a true rigid body
motion. In fact, since the translation of the nodes involved is the same for all nodes, the set of nodes
as a whole may not undergo any (rigid-body like) rotation, unlike a true rigid body. This directive,
therefore, is not a replacement for a full-fledged rigid body simulation (including rotations) in EPX.
However, it can be useful in some simplified analyses if for some reason rotations are inhibited in the
real problem.

Like several other types of links in EPX, the constraint can be imposed either in a coupled (LINK
COUP) or in a decoupled (LINK DECO) manner. The first form uses an implicit Lagrange multipliers
formulation, while the second form uses an explicit formulation, whereby the reaction forces are
computed directly. The syntax of the RIGI command is

RIGI <CENT /LECT/>

LIST /LECT/

<VECT <VX vx> <VY vy> <VZ vz>>

and produces the automatic prescription of mechanical relations (links) such that:

1. either the displacement of each point in a certain set of points equals the displacement of a
reference point C, like if all these points were all rigidly connected among them;

2. or, the displacements of each point in a certain set of points are all equal, like if all these points
were all rigidly connected among them.

Note that the first definition requires the choice of a reference node (CENT) while in the second one
no reference node is indicated and all nodes in LIST play the same role. The two alternative definitions
given above are logically equivalent, but they lead to two different forms of the links matrix. It was
found by practical experimentation that in some applications (where the number of points to be rigidly
linked is very high) the second form is much more efficient computationally, as far as the solution of
the links system is concerned. If the decoupled form of the link is chosen (DECO), which is the case of
interest here, then only the second form of the directive is possible.

The (optional) keyword VECT introduces the optional definition of a direction (vector) along which
the constraint will act. By default, the constraint acts along all space directions. The generated
constraints are as follows.

2.1 Form with reference node

Let us define N as the number of nodes in the LIST sub-directive. Consider the first form of the
directive (CENT has been specified.) By default (no VECT specified) this directive imposes the following
set of N (vectorial) conditions on nodal velocities v:

vC − v1 = 0

vC − v2 = 0

· · ·
vC − vN = 0

(1)

which corresponds to the following 2N or 3N scalar independent links:

vCx − v1x = 0

vCy − v1y = 0

vCz − v1z = 0 (3D only)

· · ·
vCx − vNx = 0

vCy − vNy = 0

vCz − vNz = 0 (3D only)

(2)

where C is the “central” node defined by CENT and 1, 2, · · · , N are the N nodes defined by LIST.
When a vector V is specified by VECT, then the following N conditions on nodal velocities are

imposed:

vC · V − v1 · V = 0

vC · V − v2 · V = 0

· · ·
vC · V − vN · V = 0

(3)

which correspond to the following N scalar links (assuming a 2D case):

vCxVx + vCyVy − v1xVx − v1yVy = 0

vCxVx + vCyVy − v2xVx − v2yVy = 0

· · ·
vCxVx + vCyVy − vNxVx − vNyVy = 0

(4)

2.2 Form without reference node

In the second form of the directive (CENT has not been specified), when no VECT is specified this
directive imposes the following set of N (vectorial) conditions on nodal velocities v:

v1 − v2 = 0

v2 − v3 = 0

· · ·
vN−1 − vN = 0

vN − v1 = 0

(5)

When a vector V is specified by VECT, then the following N conditions on nodal velocities are
imposed:

v1 · V − v2 · V = 0

v2 · V − v3 · V = 0

· · ·
vN−1 · V − vN · V = 0

vN · V − v1 · V = 0

(6)

The reason why the solution of the linear system of constraints may become (very) slow for large
N in the first form of the equations is that the same node C appears in all equations. The bandwidth
of the constraints matrix might become quite large.

The second (circular) form of the equations is computationally more efficient because the nodes
involved keep changing from an equation to the other (each node appearing only in two of the vector
constraint equations), so that the bandwidth of the constraints matrix can be made small upon
proper renumbering of the links and the solution becomes (much) more efficient, at least in sequential
execution.

However, it should also be noted that the circular formulation of the link only acts on the filling of
the constraints matrix. The filling of the condensed operator used to compute the Lagrange multipliers
is the same (full matrix for each space direction) in both cases. In MPI execution, severe performances
issues are observed. This is the motivation for the DECO version of this link.

3 Implementation

3.1 Sequential version

Like for many other models, the RIGI directive has been first implemented in the sequential version
of EPX, i.e. without any domain decomposition. The model basically consists of a fortran 90 module
M LINK RIGI that contains a simple data structure (used only in the decoupled case):

. . .

TYPE LINK_RIGI_DECO

INTEGER :: N_NODES ! N. of linked nodes

INTEGER, POINTER :: NODES(:)

LOGICAL :: HAS_VECTOR ! Does link act only along a given vector?

REAL(8) :: VEC(3) ! Vector along which the link acts (if any)

END TYPE LINK_RIGI_DECO

*

INTEGER :: N_LINK_RIGI_DECO ! N. of links present in the model

TYPE(LINK_RIGI_DECO), POINTER :: LINKS_RIGI_DECO(:)

. . .

plus some routines, among which the most important ones are the routine to read the directive from
the input file (READ RIGI) and the routine to solve the imposed links (SOLVE RIGI DECO) when the
decoupled form of the link is chosen, which is described below in Section 3.1.1. Note that, when the
coupled form is selected, no specific solving routine is necessary since in that case the links (eqs. 1–6)
are solved by the general coupled links formulation of EPX.

The above listed data structure is not necessary in the coupled case because in that case the
links (eqs. 1–6) are directly created by the syntax reading routine (READ RIGI) in the form of generic
“relations” (RELA) and are added to the system of (coupled) links of the problem. In the decoupled
case, a derived type LINK RIGI DECO is created that contains all the information related to one of
these links: the number of nodes (N NODES), the list of the nodes (NODES) and the optional direction
vector (HAS VECTOR and VECTOR.)

The set of specified decoupled RIGI links (if any) is stored in an array LINKS RIGI DECO, of
length N LINK RIGI DECO. The data structure for decoupled RIGI links must be suitably initialized,
saved and restored in case of splitting of the calculation, and destroyed at the end of a calculation.
These tasks are performed by routines INIT RIGI DECO, SAVE RIGI DECO, RESTORE RIGI DECO and
DESTROY RIGI DECO, respectively.

3.1.1 Solving the decoupled form of the constraint

The procedure for solving the decoupled form of a RIGI link is as follows. Let I = 1, · · · , N be the
nodes concerned by the link. First we compute the sum of the internal forces, external forces and
masses of the involved nodes (like if they would be a single rigid body):

F int
tot =

N∑
I=1

F int
I F ext

tot =
N∑
I=1

F ext
I Mtot =

N∑
I=1

MI (7)

4



Note that in the above (vectorial) equations only the translational components of the forces and
masses are considered, i.e. the first d components, where d is the space dimension (2 or 3.) Next, the
common acceleration of the nodes is given by:

a =
F ext
tot − F int

tot

Mtot
(8)

If no direction (vector) has been specified, the link acts along all spatial directions. We compute
the (decoupled) reaction force F dec

I to be applied to each node I, such that the node will have the
acceleration a:

F dec
I = MIa− F ext

I + F int
I (9)

and add it to the other external forces:

F ext
I ← F ext

I + F dec
I (10)

If a direction vector V has been specified, we compute the common normal acceleration aV , i.e.
the acceleration along the given direction:

aV = a · V (11)

and then, for each node, the normal reaction force:

F dec
In = MIaV −

(
F ext
I − F int

I

)
· V (12)

where MI is the scalar (translational) mass of the node (that we assume to be identical along all
spatial dimensions, so one can take e.g. MI = MIx.) Note that expression (12) is a scalar, while the
reaction vector is given by:

F dec
In = F dec

In V (13)

In all above treatments we have tacitly assumed that the V vector, if present, is a unit vector
(||V || = 1.) However, the user may specify a vector of any (non-zero) length, since EPX normalizes
it internally after reading the input. Finally, like in the case without a specified direction, we add the
reaction force (13) to the other external forces:

F ext
I ← F ext

I + F dec
In (14)

3.2 MPI version

Porting the RIGI model to MPI (domain decomposition) did not require any particular development
for the coupled version of the link, since this is based on the general (coupled) links data structure of
EPX, that is already fully compatible with MPI.

For the decoupled version, however, some development was necessary. The data structure specific
to LINK DECO RIGI described in Section 2 had to be extended to the multi-domain case and the routine
SOLVE RIGI DECO implementing the direct solution algorithm of Section 3.1.1 had to be generalized
so as to include also the case with several sub-domains.

The complete set of modifications is presented and shortly discussed hereafter. This may be used
as a guide for the porting to MPI of other similar decoupled link directives, or as a starting point for
the parallelization of more complex models in EPX.

The first operation to do is adding the specific data structure of the decoupled RIGI links, i.e. the
array LINKS RIGI DECO, to the sub-domains data structure, that is to the TYPE DOMAINE in module
M DOMAINE. In this way, each sub-domain will have its own copy of the list of decoupled rigid links
(suitably adapted as detailed in Section 3.2.6 below), i.e. its own copy of the LINKS RIGI DECO array.
Each rigid link in the list for any given sub-domain will contain only the nodes (if any) that belong to
this sub-domain, either internal or on the interface with other neighbour sub-domains. Therefore, it

is possible that a sub-domain’s rigid link be empty (no nodes), if the link involves no nodes belonging
to this particular sub-domain. The length of the list, i.e. the quantity N LINK RIGI DECO, is a global
quantity (the same for all sub-domains) and needs not be duplicated.

But before doing that, it is necessary to split the former M LINK RIGI module, which contained both
the data and the procedures (subroutines), into two modules, one containing only the data (a newly
created module M LINK RIGI DECO DATA) and the other containing only the procedures (M LINK RIGI.)
This distinction is always advisable, and is necessary in this case in order to avoid cycling in the USE

of modules that would instead result from the subsequent modifications.
Furthermore, the subroutine SOLVE RIGI DECO is extracted from the previous module M LINK RIGI

and placed in a separate program file. This was done before splitting the module M LINK RIGI module,
in order to avoid compilation errors due to the argument checking that occurs when a subroutine is
part of a module, since the routine was going to be called from both the sequential and the MPI parts
of the code (with appropriate arguments in each case.) After splitting the module, the subroutine
SOLVE RIGI DECO could also have been left into the “procedures” part, that is into the new version of
the module M LINK RIGI.

3.2.1 Module M LINK RIGI DECO DATA

The new data-only module M LINK RIGI DECO DATA, resulting from splitting of the former M LINK RIGI

module, reads:

MODULE M_LINK_RIGI_DECO_DATA

*

* data related to the link rigi condition (decoupled version only)

*

IMPLICIT NONE

*

SAVE

PRIVATE

*

PUBLIC :: LINK_RIGI_DECO,

> N_LINK_RIGI_DECO,

> LINKS_RIGI_DECO

*

TYPE LINK_RIGI_DECO

INTEGER :: N_NODES ! N. of linked nodes

INTEGER, POINTER :: NODES(:)

LOGICAL :: HAS_VECTOR ! Does link act only along a given vector?

REAL(8) :: VEC(3) ! Vector along which the link acts (if any)

END TYPE LINK_RIGI_DECO

*

INTEGER :: N_LINK_RIGI_DECO ! N. of links present in the model

TYPE(LINK_RIGI_DECO), POINTER :: LINKS_RIGI_DECO(:)

*

END MODULE M_LINK_RIGI_DECO_DATA

3.2.2 Module M LINK RIGI

The procedures-only module M LINK RIGI reads:

MODULE M_LINK_RIGI

*

* data and procedures related to the link rigi condition

*

USE M_ALLOCATION

*

USE M_LIAISON

USE M_LINKS_ADD

USE M_LINK_RIGI_DECO_DATA ! data for the deco version of this link

*

INCLUDE ’NONE.INC’

*

INCLUDE ’POUBTX.INC’

*

SAVE

*

PRIVATE

*

* public procedures

PUBLIC :: READ_RIGI,

> INIT_RIGI_DECO,

> SAVE_RIGI_DECO,

> RESTORE_RIGI_DECO,

> DESTROY_RIGI_DECO

*

* module name for save/restore operations

CHARACTER(16), PARAMETER :: NOMMODULE = ’M_LINK_RIGI’

*

CONTAINS

. . .

. . . (procedures not listed for brevity)

. . .

END MODULE M_LINK_RIGI

The USE of the new data module M LINK RIGI DECO DATA has been added, as highlighted in red
above. Furthermore, the subroutine SOLVE RIGI DECO has been removed from the module and is now
placed in a separate file.

3.2.3 Subroutines CALCUL and TLOOPP

In the CALCUL subroutine, the former USE statement (in CALCUL LINK DECO):

USE M_LINK_RIGI, ONLY: N_LINK_RIGI_DECO, SOLVE_RIGI_DECO

is replaced by:

USE M_LINK_RIGI_DECO_DATA, ONLY: N_LINK_RIGI_DECO

since the routine SOLVE RIGI DECO is now in a stand-alone file. Similarly, in the TLOOPP routine, the
statement:

USE M_LINK_RIGI

is replaced by:

USE M_LINK_RIGI_DECO_DATA

3.2.4 Module M DOMAINE

In the module M DOMAINE we add the following statements (in red):

MODULE M_DOMAINE

. . .

USE M_LINK_RIGI_DECO_DATA

. . .

TYPE DOMAINE

. . .

* link rigi deco variables

TYPE(LINK_RIGI_DECO), POINTER :: LINKS_RIGI_DECO(:)

!

END TYPE DOMAINE

. . .

SUBROUTINE CREE_DOMAINE

. . .

NULLIFY(DOM_CUR%LINKS_RIGI_DECO)

. . .

END SUBROUTINE CREE_DOMAINE

. . .

SUBROUTINE DEL_DOMAINE

. . .

DO IDOMAIN = 1, NBDOMAINES

DOM_CUR => DOMAIN(IDOMAIN)

. . .

* 40 rigi deco

IF (ASSOCIATED (DOM_CUR%LINKS_RIGI_DECO)) THEN

DO I = 1, N_LINK_RIGI_DECO

DEALLOCATE (DOM_CUR%LINKS_RIGI_DECO(I)%NODES)

ENDDO

DEALLOCATE (DOM_CUR%LINKS_RIGI_DECO)

N_LINK_RIGI_DECO = 0

ENDIF

*

ENDDO

. . .

END SUBROUTINE DEL_DOMAINE

. . .

END MODULE M_DOMAINE

An array LINKS RIGI DECO of type LINK RIGI DECO is added to the derived type DOMAINE, and
so becomes part of each sub-domain’s data structure. The array is nullified in the subroutine
CREE DOMAINE. Then, in the subroutine DEL DOMAINE, which takes care of the disposal of the data
structure of each sub-domain (in a loop over all sub-domains), it is necessary to add the de-allocation
of the data structure (note that also the global counter N LINK RIGI DECO is reset to zero.)

3.2.5 Subroutine DOMDEC SPLIT DATA

The subroutine DOMDEC SPLIT DATA splits the global data structure into the various sub-domains. We
add the following statements (in red) in order to split the data structure of interest here:

SUBROUTINE DOMDEC_SPLIT_DATA ( . . . )

. . .

USE M_LINK_RIGI_DECO_DATA

. . .

!

IF (N_LINK_RIGI_DECO > 0) THEN

CALL D_RIGI_DECO_SPLIT (IMPI)

CALL PRINT_CPU_CLOCK (2, ’SPLIT RIGI DECO’)

ENDIF

. . .

END SUBROUTINE DOMDEC_SPLIT_DATA

As it can be seen, the splitting task is actually demanded to a (new) lower-level subroutine
D RIGI DECO SPLIT. The call to PRINT CPU CLOCK is used to print on the listing the CPU time used
in this operation.

3.2.6 Subroutine D RIGI DECO SPLIT

The new subroutine D RIGI DECO SPLIT reads:

SUBROUTINE D_RIGI_DECO_SPLIT (IMPI)

*----------------------------------------------------------------------

* Split LINK RIGI DECO data structure onto sub-domains

*----------------------------------------------------------------------

USE M_DOMAINE

USE M_LINK_RIGI_DECO_DATA

USE M_LINK_RIGI ! for destroy_rigi_deco

*

IMPLICIT NONE

*

* args

INTEGER, INTENT(IN) :: IMPI

*

* locals

INTEGER :: I, J, JJ, JJ_L, NN

TYPE(LINK_RIGI_DECO), POINTER :: LRDG, LRDL

*

IF (IMPI /= 1) THEN

CALL ERRMSS (’D_RIGI_DECO_SPLIT’,

> ’LINK DECO RIGI N/A WITH SEQUENTIAL DOMAIN DECOMPOSITION’)

STOP ’D_RIGI_DECO_SPLIT: LINK DECO RIGI N/A WITH SEQ. DOM. DEC.’

ENDIF

*

* Attention: in the code below DOMAIN(1) always represents the current

* sub-domain, i.e. the sub-domain associated with the current processor in MPI,

* and not the first sub-domain of the domain decomposition !!!

* The present routine is called by each processor in MPI, not only by

* the 0-processor.

*

ALLOCATE (DOMAIN(1)%LINKS_RIGI_DECO(N_LINK_RIGI_DECO))

*

DO I = 1, N_LINK_RIGI_DECO

LRDG => LINKS_RIGI_DECO(I)

LRDL => DOMAIN(1)%LINKS_RIGI_DECO(I)

*

LRDL%HAS_VECTOR = LRDG%HAS_VECTOR

LRDL%VEC(1:3) = LRDG%VEC(1:3)

*

NN = 0

DO J = 1, LRDG%N_NODES

JJ = LRDG%NODES(J)

JJ_L = DOMAIN(1)%CONVERS%NOEUD_GL(JJ)

IF (JJ_L > 0) NN = NN + 1

ENDDO

*

LRDL%N_NODES = NN

*

IF (NN > 0) THEN ! Attention: NN can be 0 if the link involves

! no nodes belonging to the current sub-domain

ALLOCATE (LRDL%NODES(NN))

*

NN = 0

DO J = 1, LRDG%N_NODES

JJ = LRDG%NODES(J)

JJ_L = DOMAIN(1)%CONVERS%NOEUD_GL(JJ)

IF (JJ_L > 0) THEN

NN = NN + 1

LRDL%NODES(NN) = JJ_L

ENDIF

ENDDO

ELSE

NULLIFY (LRDL%NODES)

ENDIF

ENDDO

*

NN = N_LINK_RIGI_DECO ! SAVE THE NUMBER OF RIGI DECO LINKS

CALL DESTROY_RIGI_DECO ! DESTROY THE GLOBAL STRUCTURE

N_LINK_RIGI_DECO = NN ! RESET THE NUMBER

*

END SUBROUTINE D_RIGI_DECO_SPLIT

Before discussing this subroutine in detail, it is essential to clarify the meaning of DOMAIN(1) in the
above code. In fact, it should be noted that the subroutine D RIGI DECO SPLIT is called by (executed
on) each MPI process (i.e., for each sub-domain) separately, and not, as one might perhaps expect
from its name, only on the first processor and treating all sub-domains. Therefore, in the above code
DOMAIN(1) represents the current sub-domain (the one associated with the current processor), and
not the first sub-domain of the decomposition, as one might think at first sight.

It turns out in fact that on the first processor (processor 0 in MPI) all sub-domains are known
(NBDOMAINES is equal to the number of processors used in the MPI execution), so DOMAIN(1) is indeed
the current sub-domain for this processor. But, on all other processors, only the “local” sub-domain
is known (NBDOMAINES is equal to 1) so that, again, DOMAIN(1) is the current sub-domain (the sub-
domain associated with the current processor.)

It should also be mentioned in passing that the possibility of having several domains inside one
process is a result of the original implementation of domain decomposition for sequential computing.
This is why the one-dimensional array DOMAIN(:) is present in the implementation, even though
only DOMAIN(1) is basically used with MPI. By the way, even with MPI, the process with rank 0 uses
light versions of sub-domains attributed to other processes (DOMAIN(I), I > 1) for outputs gathering
purposes.

Note also that in the current implementation, a test producing an error if the IMPI variable is
not 1 has been added in the splitting routine, preventing the wrong use of RIGI DECO links with the
sequential domain decomposition.

Basically, the routine copies the global data structure to the current sub-domain (DOMAIN(1)), with
suitable adaptations. As concerns the nodes involved in each decoupled RIGI link, only the linked

nodes which belong to the current sub-domain are actually copied onto the split data-structure. A
(global) node JJ belongs to the current sub-domain (DOMAIN(1)) if its local number JJ L in this sub-
domain is non-zero. The global-to-local nodal index conversion for a sub-domain is performed by the
sub-domain’s table CONVERS%NOEUD GL, in this case DOMAIN(1)%CONVERS%NOEUD GL(JJ) for the global
node JJ.

Note that it may happen that none of the nodes of a link belong to the current domain. In this
case, the link will still be present in the current sub-domain’s list, but it will have N NODES set to 0
and NODES not allocated.

Once done the copy, the global data structure can be de-allocated by calling DESTROY RIGI DECO.
This will produce an error if any attempt is made to use the global data structure during the tran-
sient calculation (correctly, since this is forbidden.) The only item of the global data structure that
must be preserved is the total number of decoupled RIGI links N LINK RIGI DECO, because this quan-
tity will be used by each sub-domain, as already mentioned. Since the routine DESTROY RIGI DECO

resets N LINK RIGI DECO to zero, we must save the old value and then restore it after the call to
DESTROY RIGI DECO, as shown in the last part of the routine.

3.2.7 Subroutine D CALFREE

The subroutine D CALFREE computes the free accelerations for each sub-domain. This is where we
must add the call to the SOLVE RIGI DECO subroutine (in case of domain decomposition.) The added
statements are highlighted in red below:

SUBROUTINE D_CALFREE ( . . . )

. . .

USE M_LINK_RIGI_DECO_DATA

. . .

*

IF (N_LINK_RIGI_DECO > 0) THEN

* decoupled rigid connections (link deco rigi)

CALL SOLVE_RIGI_DECO (DM_FI%VAR, DM_FE%VAR, DM_FC%VAR,

& DM_XMS%VAR, DM_FI%POS)

ENDIF

. . .

END SUBROUTINE D_CALFREE

Note the variables passed in the exchange list when calling SOLVE RIGI DECO from the MPI part of
the code (i.e. from D CALFREE.) The same call from the sequential part (i.e., from CALCUL or TLOOPP)
reads:

CALL SOLVE_RIGI_DECO (FI, FE, FDECO, XM, POSP)

We see therefore that DM FI%VAR is the vector of nodal internal forces for the current sub-domain
and corresponds to the FI global array of the sequential case (without domain decomposition.) The
DM FE%VAR are the external forces for the current sub-domain and correspond to FE in sequential.
The DM FC%VAR are the complementary forces (i.e. the decoupled reaction forces in this case) for the
current sub-domain and correspond to FDECO in sequential.

The DM FI%POS array is the array of pointers into the nodal degrees of freedom for the current
sub-domain (POSP in sequential.) This array is the same for all nodal variables such as displacements,
velocities, accelerations, forces, masses etc. (but not for nodal coordinates.) Therefore one can choose
the pointer of any of these variables, in this case the internal forces, to represent POSP.

It should be noted that for nodes along the interface of the current sub-domain, i.e. for nodes that
belong to one or more other neighbouring sub-domains besides the present one, the above arrays of
forces contain the total force contributions, i.e. the forces resulting from assembly of elements of the
current as well as of neighbour sub-domains. In order to be consistent with this definition, one must
use the DM XMS%VAR array for the nodal masses, as shown above, and not the DM XM%VAR array. In
fact, the latter contains only the mass contributions from the current domain.

3.2.8 Subroutine SOLVE RIGI DECO

As already mentioned, the subroutine SOLVE RIGI DECO is extracted from the module M LINK RIGI

and placed in a stand-alone file. Furthermore, the routine is generalized so that it can be called
from either the sequential part (CALCUL or TLOOPP) or the MPI part (D CALFREE) of the code. The
treatment of the multi-domain case is highlighted in red and commented step-by-step below:

SUBROUTINE SOLVE_RIGI_DECO (FI, FE, FDECO, XM, POSP)

*

USE M_LINK_RIGI_DECO_DATA

USE M_DOMAINE

USE M_INTERFACE

USE M_DOMAINE_MPI

*

The USE of the new data-only module and of various modules related to sub-domains is added, in
order to access the necessary quantities.

INCLUDE "NONE.INC"

*

INCLUDE "CONTRO.INC" ! FOR IDIM

*

* args

REAL(8), INTENT(IN) :: FI(*), XM(*)

INTEGER, INTENT(IN) :: POSP(*)

REAL(8), INTENT(INOUT) :: FE(*), FDECO(*)

*

* locals

INTEGER :: I, K, IAD, NOD, NODG, NN

TYPE(LINK_RIGI_DECO), POINTER :: LRD

REAL(8) :: FISUM(3), FESUM(3), XMSUM(3), ACC(3), ACCN, RN,

& RBUF(9)

REAL(8), ALLOCATABLE :: WGT(:)

*

Several local variables are added, in particular the RBUF array that will be used to exchange data
between the processes (sub-domains) via the MPI protocol. Also an allocatable array of weights WGT

is added, whose purpose (with domain decomposition only) will be described in the following.

NULLIFY (LRD)

*

DO I = 1, N_LINK_RIGI_DECO

* total forces and total mass

FISUM(:) = 0.D0

FESUM(:) = 0.D0

XMSUM(:) = 0.D0

*

IF (IDOM_CURRENT == 0) THEN

LRD => LINKS_RIGI_DECO(I)

ELSE

LRD => DOMAIN(1)%LINKS_RIGI_DECO(I)

ENDIF

NN = LRD%N_NODES ! N. of nodes involved (may be 0 in MPI)

*

The local pointer LRD to the i-th decoupled RIGI link is associated with the global array item
LINKS RIGI DECO(I) in the sequential case (IDOM CURRENT == 0), while it is associated with the
local array item of the current sub-domain (DOMAIN(1)%LINKS RIGI DECO(I)) in the MPI case. The
number of nodes involved by the present link is named NN for convenience. Note that NN may be 0 in
MPI calculations, as already observed previously.

IF (IDOM_CURRENT > 0) THEN ! Compute the nodal weights

IF (NN > 0) THEN

ALLOCATE (WGT(NN))

DO K = 1, LRD%N_NODES

NOD = LRD%NODES(K)

NODG = DOMAIN(1)%CONVERS%NOEUD_LG(NOD)

WGT(K) = 1.D0 / DBLE (CORRESP(0,NODG)) ! 1 / Number of sub-domains

! containing NODG

ENDDO

ENDIF

ENDIF

*

If domain decomposition is active and the number of linked nodes NN is greater than zero, then the
nodal weights array WGT is allocated for each of the involved nodes and then filled up. In the sequential
case weights are not needed. In the MPI case, the weight is set to the inverse of the number of sub-
domains which actually contain the node under consideration. First, the local node index NOD is taken
from the LRD%NODES array (of the first sub-domain.) Then the node index is converted from local to
global (NODG) by means of the DOMAIN(1)%CONVERS%NOEUD LG(NOD). Next, the number of sub-domains
that include the (global) node (NODG) is returned from the INTEGER function CORRESP(0,NODG) (in
M INTERFACE.) Finally, the nodal weight is set to the inverse of the above number.

IF (IDOM_CURRENT == 0) THEN

DO K = 1, NN

NOD = LRD%NODES(K)

IAD = POSP(NOD) - 1

FISUM(1:IDIM) = FISUM(1:IDIM) + FI(IAD+1:IAD+IDIM)

FESUM(1:IDIM) = FESUM(1:IDIM) + FE(IAD+1:IAD+IDIM)

XMSUM(1:IDIM) = XMSUM(1:IDIM) + XM(IAD+1:IAD+IDIM)

END DO

ELSE

IF (NN > 0) THEN

DO K = 1, NN

NOD = LRD%NODES(K)

IAD = POSP(NOD) - 1

FISUM(1:IDIM) = FISUM(1:IDIM) + FI(IAD+1:IAD+IDIM)*WGT(K)

FESUM(1:IDIM) = FESUM(1:IDIM) + FE(IAD+1:IAD+IDIM)*WGT(K)

XMSUM(1:IDIM) = XMSUM(1:IDIM) + XM(IAD+1:IAD+IDIM)*WGT(K)

END DO

ENDIF

*

RBUF(1:9) = 0.D0

RBUF(1:IDIM) = FISUM(1:IDIM)

RBUF(4:3+IDIM) = FESUM(1:IDIM)

RBUF(7:6+IDIM) = XMSUM(1:IDIM)

*

CALL FMPI_RALLSUM_IN_PLACE (RBUF, 9)

*

FISUM(1:IDIM) = RBUF(1:IDIM)

FESUM(1:IDIM) = RBUF(4:3+IDIM)

XMSUM(1:IDIM) = RBUF(7:6+IDIM)

ENDIF

*

* common acceleration

ACC(1:IDIM) = (FESUM(1:IDIM) - FISUM(1:IDIM)) / XMSUM(1:IDIM)

*

Now the common acceleration has to be computed. The internal forces, external forces, and masses
of all nodes concerned by the current link which belong to the current sub-domain are summed up into
the FISUM, FESUM and XMSUM arrays (of dimension d, the space dimension), respectively. In the MPI
case, each term in the summations is multiplied by the weight of the node concerned. Thus, if a node
belongs to more than one sub-domain, this scaling will ensure that the total sum is correct.

Then, only in the MPI case, the contributions of all other sub-domains are included in the sum-
mations. In the present case, this is simply accomplished by calling the FMPI RALLSUM IN PLACE

subroutine, which takes care of the necessary communications between the processes. Note, however,
that the present one is a particularly simple model so that the data exchanges (gathering and scat-
tering) for the actual computation are very simple. But they could be rather more complicated in a
more generic case (i.e., for a more general model.)

First, the contents of the three sums (for the current sub-domain only) are inserted in the array
RBUF, which has a length of 9. Then the FMPI RALLSUM IN PLACE subroutine is called, by passing
to it the array (RBUF) and its length (9.) Upon return from the subroutine, the array contains the
complete sums over all sub-domains. These values are copied back into the arrays FISUM, FESUM and
XMSUM, ready for the calculation of the common acceleration ACC, which is performed next. Finally,
we can compute the reactions, as shown next.

* reactions

IF (.NOT. LRD%HAS_VECTOR) THEN

* no vector was specified for this link deco rigi

IF (IDOM_CURRENT == 0) THEN

DO K = 1, NN

NOD = LRD%NODES(K)

IAD = POSP(NOD) - 1

FDECO(IAD+1:IAD+IDIM) = XM(IAD+1:IAD+IDIM)*ACC(1:IDIM)

> - FE(IAD+1:IAD+IDIM)

> + FI(IAD+1:IAD+IDIM)

FE(IAD+1:IAD+IDIM) = FE(IAD+1:IAD+IDIM) +

> FDECO(IAD+1:IAD+IDIM)

END DO

ELSE

IF (NN > 0) THEN

DO K = 1, NN

NOD = LRD%NODES(K)

IAD = POSP(NOD) - 1

FDECO(IAD+1:IAD+IDIM) = (XM(IAD+1:IAD+IDIM)*ACC(1:IDIM)

> - FE(IAD+1:IAD+IDIM)

> + FI(IAD+1:IAD+IDIM))*WGT(K)

END DO

ENDIF

ENDIF

We consider first the case that no direction (vector) has been specified for the current link, so
that the link will act along all spatial directions. The difference in the MPI case with respect to the
sequential case is that the computed reaction force must be multiplied by the (nodal) weight, in order
to account for the fact that a node may belong to more than one sub-domain.

Another notable difference between the two cases is that in the sequential case the newly computed
reactions (FDECO) are added also to the corresponding external forces (FE), while this operation must
not be done in the MPI case, since in MPI the treatment of external forces is different.

In fact, in MPI the decoupled link forces must be first transferred to remote nodes (if any), for
links based on spatial proximity such as contact or FLSR, and then summed over the interfaces of the
sub-domains to obtain the correct vector to be added to the FE vector. This is why in MPI they must
not be added immediately to FE and included only into FC (FDECO here), to which the generic parallel
treatment will then be applied.

ELSE

* vector was specified for this link deco rigi

ACCN = ACC(1)*LRD%VEC(1) + ACC(2)*LRD%VEC(2)

IF (IDIM > 2) THEN

ACCN = ACCN + ACC(3)*LRD%VEC(3)

ENDIF

IF (IDOM_CURRENT == 0) THEN

DO K = 1, NN

NOD = LRD%NODES(K)

IAD = POSP(NOD) - 1

* rn is the reaction along the normal (i.e. the specifed vector)

RN = ACCN*XM(IAD+1) - ! Assume xm(iad+1:iad+idim) are equal !!!

> (FE(IAD+1) - FI(IAD+1))*LRD%VEC(1) -

> (FE(IAD+2) - FI(IAD+2))*LRD%VEC(2)

IF (IDIM > 2) THEN

RN = RN - (FE(IAD+3) - FI(IAD+3))*LRD%VEC(3)

ENDIF

FDECO(IAD+1:IAD+IDIM) = RN*LRD%VEC(1:IDIM)

FE(IAD+1:IAD+IDIM) = FE(IAD+1:IAD+IDIM) +

> FDECO(IAD+1:IAD+IDIM)

END DO

ELSE

IF (NN > 0) THEN

DO K = 1, NN

NOD = LRD%NODES(K)

IAD = POSP(NOD) - 1

* rn is the reaction along the normal (i.e. the specifed vector)

RN = ACCN*XM(IAD+1) - ! Assume xm(iad+1:iad+idim) are equal !!!

> (FE(IAD+1) - FI(IAD+1))*LRD%VEC(1) -

> (FE(IAD+2) - FI(IAD+2))*LRD%VEC(2)

IF (IDIM > 2) THEN

RN = RN - (FE(IAD+3) - FI(IAD+3))*LRD%VEC(3)

ENDIF

RN = RN*WGT(K)

FDECO(IAD+1:IAD+IDIM) = RN*LRD%VEC(1:IDIM)

END DO

ENDIF

ENDIF

ENDIF

*

IF (IDOM_CURRENT > 0) THEN

IF (NN > 0) DEALLOCATE (WGT)

ENDIF

*

END DO ! I = 1, N_LINK_RIGI_DECO

*

END SUBROUTINE SOLVE_RIGI_DECO

The case of a link with a specified direction (vector), listed above, is similar to the case with
no vector as far as the distinction between sequential and MPI cases is concerned, and deserves no
further comment. The weights array WGT is de-allocated at the end of the treatment of each link if it
had been previously allocated, i.e. if MPI is active and the number of involved nodes is greater than
zero.

4 Numerical examples

We first consider some numerical examples using the sequential version of EPX (as a reference.) Then,
(conceptually) the same examples are run by the MPI version. All examples are taken from the EPX
non-regression series of benchmark tests.

4.1 Sequential examples

The sequential examples are listed in Table 1. Both calculations are contained in the benchmark test
BM STR RIGI DECO.

Test name Mesh Description Final time [ms] Steps CPU [s]

DRIG01 2 Q41L Without vector 10.0 47 0.03
DRIG02 2 Q41L With vector 10.0 47 0.03

Table 1: Numerical simulations with the sequential version of EPX.

4.1.1 Case DRIG01

The geometry of the problem is illustrated in Fig. 1(a). A 2D rectangle is discretized in two square
elements of type Q41L. An external force, constant in time and directed along the positive x-axis is
applied to node 1 (in the left-bottom corner of the model.) The RIGI condition is applied to all nodes:

LINK DECO

RIGI LIST LECT tous TERM

and therefore the body is expected to translate (without rotating) along the positive x-direction like
if it were rigid, i.e. with all nodes having the same displacement. This behaviour is already confirmed
by Fig. 1(b), where we see that the initial acceleration is the same at all nodes, not only at the
loaded one. Note, incidentally, that it is not possible to draw (only) the applied force in Fig. 1(a) by
visualizing the external forces, since these include also the reactions.

The good quality of the solution is confirmed in Fig. 2, where on the left we see that all x-
displacements are equal and parabolic in time, while on the right we see that all y-displacements are
zero.

(a) Geometry (b) Initial accelerations

Figure 1: Test case DRIG01.

Figure 2: Results of test case DRIG01.

4.1.2 Case DRIG02

This test uses the same geometry as case DRIG01 but uses the vector definition in the RIGI condition.
The loading is similar, but multiplied by a factor 100 in order to increase visibility of results. The
boundary conditions are as follows. The nodes on the right (nodes 2, 4 and 6 in Fig. 1(a)) are blocked
horizontally. A LINK DECO RIGI constraint is applied to the nodes on the left (nodes 1, 3 and 5 in
Fig. 1(a)) and a vector direction is specified:

LINK DECO

RIGI LIST LECT 1 3 5 TERM

VECT VX 1.0 VY 0.0

BLOQ 1 LECT 2 4 6 TERM

Since the vector direction is along x, aligned with the force, it should have no great effect on the
solution, except for the fact that, unlike in the previous test DRIG01, here the nodes are left free of
having different displacements along y, which indeed should occur due to the Poisson effect (ν = 0.3.)

From Fig. 3(a) we see that, as expected, the left nodes have equal (and oscillating) x-displacements,
while the right nodes have zero x-displacements. From Fig. 3(b) we see that, also as expected, the
lower and the upper nodes have equal and opposite (oscillating) y-displacements, while the middle
nodes have zero y-displacements.

Figure 3: Results of test case DRIG02.

4.2 MPI examples

The MPI examples are listed in Table 2. The first calculation (DRIG03) is contained in the benchmark
test BM MPI STR RIGI DECO, while the second calculation (DRIG04) is contained in test BM MPI STR-

RIGI DECO VECT. This is due to the fact that it is currently not possible to perform more than one
calculation in the same MPI benchmark (but it is possible to perform one calculation followed by the
reading of its results via RESU and the plotting of PostScript curves.)

Test name Mesh Description Final time [ms] Steps CPU [s]

DRIG03 10 Q41L Without vector 10.0 47 0.3
DRIG04 10 Q41L With vector 10.0 47 0.3

Table 2: Numerical simulations with the MPI version of EPX (with 4 processes.)

4.2.1 Case DRIG03

This test is similar to the sequential case DRIG01, but the number of elements has been increased
from 2 to 10 as shown in Fig. 4(a), in order to be able to run the test under MPI on 4 processors,
since the number of processors must be smaller than or equal to the number of elements.

The same force as in case DRIG01 is applied, therefore the acceleration should be 5 times smaller
since the mass of the model is 5 times bigger here. The factor 5 is indeed confirmed by comparing
Figs. 1(b) and 4(b). Note also that the CHAR FCTE (constant force) directive is at the moment
incompatible with the MPI version of EPX, so it has been replaced by the CHAR FACT FORC directive,
by using a constant table in time. Furthermore, the writing of a log file (OPTI LOG) has been disabled
because this is believed to penalize MPI calculations (although in this simple case this is of course
completely irrelevant.)

The good quality of the solution is confirmed in Fig. 5, where on the left we see that all x-
displacements are equal and parabolic in time, while on the right we see that all y-displacements are
zero.

(a) Geometry (b) Initial accelerations

Figure 4: Test case DRIG03.

Figure 5: Results of test case DRIG03 (on 4 processors.)

4.2.2 Case DRIG04

This test is similar to the sequential case DRIG02 (with a vector specified), but it uses the same
10-element mesh as case DRIG03. The boundary conditions and the RIGI directive are set similarly
to case DRIG02.

From Fig. 3(a) we see that, as expected, the left nodes have equal (and oscillating) x-displacements,
while the right nodes have zero x-displacements. From Fig. 3(b) we see that, also as expected, the
lower and the upper nodes have equal and opposite (oscillating) y-displacements.

Figure 6: Results of test case DRIG04 (on 4 processors.)

Appendix I — Input files

All the input files used in the previous Sections are listed below.

bm mpi str rigi deco.epx

$ BM_MPI_STR_RIGI_DECO TOUS MLARCHER 17/08/29 22:32:50 #3326

DRIG03

ECHO

BMPI

!CONV WIN

LAGR CPLA

GEOM LIBR POIN 22 Q41L 10 TERM

0 0 1 0

0 1 1 1

0 2 1 2

0 3 1 3

0 4 1 4

0 5 1 5

0 6 1 6

0 7 1 7

0 8 1 8

0 9 1 9

0 10 1 10

1 2 4 3

3 4 6 5

5 6 8 7

7 8 10 9

9 10 12 11

11 12 14 13

13 14 16 15

15 16 18 17

17 18 20 19

19 20 22 21

EPAI 1. LECT 1 PAS 1 10 TERM

MATE VM23 RO 8000. YOUN 1.D11 NU 0.3 ELAS 2.D8

TRAC 3 2.D8 2.D-3 3.D8 1. 3.1D8 2.

LECT 1 PAS 1 10 TERM

CHAR 1 FACT 2

FORC 1 1.E5 LECT 1 TERM

TABL 2 0. 1.

1. 1.

LINK DECO

RIGI LIST LECT tous TERM

ECRI DEPL VITE ACCE FINT FEXT FLIA FDEC CONT ECRO FREQ 100

FICH ALIC FREQ 1

FICH PVTK FREQ 1

GROUP AUTO

VARI DEPL VITE ACCE ECRO

OPTI PAS AUTO NOTE ! LOG 1

CALC TINI 0. TEND 1.D-2

*=================================================================

SUIT

Post-treatment (time curves from alice file)

ECHO

RESU ALIC GARD PSCR

SORT GRAP

AXTE 1.0 ’Time [s]’

COUR 1 ’dx_1’ DEPL COMP 1 NOEU LECT 1 TERM

COUR 2 ’dx_2’ DEPL COMP 1 NOEU LECT 2 TERM

COUR 3 ’dx_21’ DEPL COMP 1 NOEU LECT 21 TERM

COUR 4 ’dx_22’ DEPL COMP 1 NOEU LECT 22 TERM

COUR 11 ’dy_1’ DEPL COMP 2 NOEU LECT 1 TERM

COUR 12 ’dy_2’ DEPL COMP 2 NOEU LECT 2 TERM

COUR 13 ’dy_21’ DEPL COMP 2 NOEU LECT 21 TERM

COUR 14 ’dy_22’ DEPL COMP 2 NOEU LECT 22 TERM

COUR 21 ’sx_1’ CONT COMP 1 ELEM LECT 1 TERM

COUR 22 ’sy_1’ CONT COMP 2 ELEM LECT 1 TERM

COUR 23 ’ex_1’ EPST COMP 1 ELEM LECT 1 TERM

COUR 24 ’ey_1’ EPST COMP 2 ELEM LECT 1 TERM

TRAC 1 2 3 4 AXES 1.0 ’DEPL. [M]’ YZER

TRAC 11 12 13 14 AXES 1.0 ’DEPL. [M]’ YZER

TRAC 21 22 AXES 1.0 ’CONT. [PA]’ YZER

TRAC 23 24 AXES 1.0 ’DEFO. [-]’ YZER

*=================================================================

QUAL DEPL COMP 1 LECT 1 TERM REFE 6.43261E-5 TOLE 1.E-3

DEPL COMP 1 LECT 22 TERM REFE 6.43261E-5 TOLE 1.E-3

FIN

BEGIN DESCRIPTION

This bench checks the decoupled form of the RIGI link

(LINK DECO RIGI) under MPI.

END DESCRIPTION

bm mpi str rigi deco vect.epx

$ BM_MPI_STR_RIGI_DECO_VECT TOUS MLARCHER 17/08/29 22:32:50 #3326

DRIG04

ECHO

BMPI

!CONV WIN

LAGR CPLA

GEOM LIBR POIN 22 Q41L 10 TERM

0 0 1 0

0 1 1 1

0 2 1 2

0 3 1 3

0 4 1 4

0 5 1 5

0 6 1 6

0 7 1 7

0 8 1 8

0 9 1 9

0 10 1 10

1 2 4 3

3 4 6 5

5 6 8 7

7 8 10 9

9 10 12 11

11 12 14 13

13 14 16 15

15 16 18 17

17 18 20 19

19 20 22 21

EPAI 1. LECT 1 PAS 1 10 TERM

MATE VM23 RO 8000. YOUN 1.D11 NU 0.3 ELAS 2.D8

TRAC 3 2.D8 2.D-3 3.D8 1. 3.1D8 2.

LECT 1 PAS 1 10 TERM

CHAR 1 FACT 2

FORC 1 1.E7 LECT 1 TERM

TABL 2 0. 1.

1. 1.

LINK COUP ! link deco bloq is not available under MPI ...

BLOQ 1 LECT 2 PAS 2 22 TERM

LINK DECO

RIGI LIST LECT 1 PAS 2 21 TERM

VECT VX 1.0 VY 0.0

ECRI DEPL VITE ACCE FINT FEXT FLIA FDEC CONT ECRO FREQ 100

FICH ALIC FREQ 1

FICH PVTK FREQ 1

GROUP AUTO

VARI DEPL VITE ACCE ECRO

OPTI PAS AUTO NOTE ! LOG 1

CALC TINI 0. TEND 1.D-2

*=================================================================

SUIT

Post-treatment (time curves from alice file)

ECHO

RESU ALIC GARD PSCR

SORT GRAP

AXTE 1.0 ’Time [s]’

COUR 1 ’dx_1’ DEPL COMP 1 NOEU LECT 1 TERM

COUR 2 ’dx_2’ DEPL COMP 1 NOEU LECT 2 TERM

COUR 3 ’dx_21’ DEPL COMP 1 NOEU LECT 21 TERM

COUR 4 ’dx_22’ DEPL COMP 1 NOEU LECT 22 TERM

COUR 11 ’dy_1’ DEPL COMP 2 NOEU LECT 1 TERM

COUR 12 ’dy_2’ DEPL COMP 2 NOEU LECT 2 TERM

COUR 13 ’dy_21’ DEPL COMP 2 NOEU LECT 21 TERM

COUR 14 ’dy_22’ DEPL COMP 2 NOEU LECT 22 TERM

COUR 21 ’sx_1’ CONT COMP 1 ELEM LECT 1 TERM

COUR 22 ’sy_1’ CONT COMP 2 ELEM LECT 1 TERM

COUR 23 ’ex_1’ EPST COMP 1 ELEM LECT 1 TERM

COUR 24 ’ey_1’ EPST COMP 2 ELEM LECT 1 TERM

TRAC 1 2 3 4 AXES 1.0 ’DEPL. [M]’ YZER

TRAC 11 12 13 14 AXES 1.0 ’DEPL. [M]’ YZER

TRAC 21 22 AXES 1.0 ’CONT. [PA]’ YZER

TRAC 23 24 AXES 1.0 ’DEFO. [-]’ YZER

*=================================================================

QUAL DEPL COMP 1 LECT 1 TERM REFE 1.28801E-5 TOLE 1.E-3

DEPL COMP 1 LECT 21 TERM REFE 1.28801E-5 TOLE 1.E-3

FIN

BEGIN DESCRIPTION

This bench checks the decoupled form of the RIGI link

(LINK DECO RIGI) with VECT under MPI.

END DESCRIPTION

bm str rigi deco.epx

$ BM_STR_RIGI_DECO TOUS GVALSAMOS 17/07/27 22:25:41 #3304

DRIG01

ECHO

!CONV WIN

LAGR CPLA

GEOM LIBR POIN 6 Q41L 2 TERM

0 0 1 0

0 1 1 1

0 2 1 2

1 2 4 3

3 4 6 5

EPAI 1. LECT 1 PAS 1 2 TERM

MATE VM23 RO 8000. YOUN 1.D11 NU 0.3 ELAS 2.D8

TRAC 3 2.D8 2.D-3 3.D8 1. 3.1D8 2.

LECT 1 2 TERM

CHAR FCTE NODE LECT 1 TERM

FORC 1.D5 VECT 1.0 0.0 0.0

LINK DECO

RIGI LIST LECT tous TERM

ECRI DEPL VITE ACCE FINT FEXT FLIA FDEC CONT ECRO FREQ 100

FICH ALIC FREQ 1

OPTI PAS AUTO NOTE LOG 1

CALC TINI 0. TEND 1.D-2

*=================================================================

SUIT

Post-treatment (time curves from alice file)

ECHO

RESU ALIC GARD PSCR

SORT GRAP

AXTE 1.0 ’Time [s]’

COUR 1 ’dx_1’ DEPL COMP 1 NOEU LECT 1 TERM

COUR 2 ’dx_2’ DEPL COMP 1 NOEU LECT 2 TERM

COUR 3 ’dx_3’ DEPL COMP 1 NOEU LECT 3 TERM

COUR 4 ’dx_4’ DEPL COMP 1 NOEU LECT 4 TERM

COUR 5 ’dx_5’ DEPL COMP 1 NOEU LECT 5 TERM

COUR 6 ’dx_6’ DEPL COMP 1 NOEU LECT 6 TERM

COUR 11 ’dy_1’ DEPL COMP 2 NOEU LECT 1 TERM

COUR 12 ’dy_2’ DEPL COMP 2 NOEU LECT 2 TERM

COUR 13 ’dy_3’ DEPL COMP 2 NOEU LECT 3 TERM

COUR 14 ’dy_4’ DEPL COMP 2 NOEU LECT 4 TERM

COUR 15 ’dy_5’ DEPL COMP 2 NOEU LECT 5 TERM

COUR 16 ’dy_6’ DEPL COMP 2 NOEU LECT 6 TERM

COUR 21 ’sx_1’ CONT COMP 1 ELEM LECT 1 TERM

COUR 22 ’sy_1’ CONT COMP 2 ELEM LECT 1 TERM

COUR 23 ’ex_1’ EPST COMP 1 ELEM LECT 1 TERM

COUR 24 ’ey_1’ EPST COMP 2 ELEM LECT 1 TERM

TRAC 1 2 3 4 5 6 AXES 1.0 ’DEPL. [M]’ YZER

TRAC 11 12 13 14 15 16 AXES 1.0 ’DEPL. [M]’ YZER

TRAC 21 22 AXES 1.0 ’CONT. [PA]’ YZER

TRAC 23 24 AXES 1.0 ’DEFO. [-]’ YZER

*=================================================================

QUAL DEPL COMP 1 LECT 1 TERM REFE 3.21630E-4 TOLE 1.E-3

DEPL COMP 1 LECT 6 TERM REFE 3.21630E-4 TOLE 1.E-3

SUIT

DRIG02

ECHO

!CONV WIN

LAGR CPLA

GEOM LIBR POIN 6 Q41L 2 TERM

0 0 1 0

0 1 1 1

0 2 1 2

1 2 4 3

3 4 6 5

EPAI 1. LECT 1 PAS 1 2 TERM

MATE VM23 RO 8000. YOUN 1.D11 NU 0.3 ELAS 2.D8

TRAC 3 2.D8 2.D-3 3.D8 1. 3.1D8 2.

LECT 1 2 TERM

CHAR FCTE NODE LECT 1 TERM

FORC 1.D7 VECT 1.0 0.0 0.0

LINK DECO

RIGI LIST LECT 1 3 5 TERM

VECT VX 1.0 VY 0.0

BLOQ 1 LECT 2 4 6 TERM

ECRI DEPL VITE ACCE FINT FEXT FLIA FDEC CONT ECRO FREQ 100

FICH ALIC FREQ 1

OPTI PAS AUTO NOTE LOG 1

CALC TINI 0. TEND 1.D-2

*=================================================================

SUIT

Post-treatment (time curves from alice file)

ECHO

RESU ALIC GARD PSCR

SORT GRAP

AXTE 1.0 ’Time [s]’

COUR 1 ’dx_1’ DEPL COMP 1 NOEU LECT 1 TERM

COUR 2 ’dx_2’ DEPL COMP 1 NOEU LECT 2 TERM

COUR 3 ’dx_3’ DEPL COMP 1 NOEU LECT 3 TERM

COUR 4 ’dx_4’ DEPL COMP 1 NOEU LECT 4 TERM

COUR 5 ’dx_5’ DEPL COMP 1 NOEU LECT 5 TERM

COUR 6 ’dx_6’ DEPL COMP 1 NOEU LECT 6 TERM

COUR 11 ’dy_1’ DEPL COMP 2 NOEU LECT 1 TERM

COUR 12 ’dy_2’ DEPL COMP 2 NOEU LECT 2 TERM

COUR 13 ’dy_3’ DEPL COMP 2 NOEU LECT 3 TERM

COUR 14 ’dy_4’ DEPL COMP 2 NOEU LECT 4 TERM

COUR 15 ’dy_5’ DEPL COMP 2 NOEU LECT 5 TERM

COUR 16 ’dy_6’ DEPL COMP 2 NOEU LECT 6 TERM

COUR 21 ’sx_1’ CONT COMP 1 ELEM LECT 1 TERM

COUR 22 ’sy_1’ CONT COMP 2 ELEM LECT 1 TERM

COUR 23 ’ex_1’ EPST COMP 1 ELEM LECT 1 TERM

COUR 24 ’ey_1’ EPST COMP 2 ELEM LECT 1 TERM

TRAC 1 2 3 4 5 6 AXES 1.0 ’DEPL. [M]’ YZER

TRAC 11 12 13 14 15 16 AXES 1.0 ’DEPL. [M]’ YZER

TRAC 21 22 AXES 1.0 ’CONT. [PA]’ YZER

TRAC 23 24 AXES 1.0 ’DEFO. [-]’ YZER

QUAL DEPL COMP 1 LECT 1 TERM REFE 9.05053E-5 TOLE 1.E-3

DEPL COMP 1 LECT 5 TERM REFE 9.05053E-5 TOLE 1.E-3

*=================================================================

FIN

BEGIN DESCRIPTION

This bench checks the decoupled form of the RIGI link

(LINK DECO RIGI).

There are two calculations:

1) DRIG01 : form without VECT.

2) DRIG02 : form with VECT.

END DESCRIPTION

post01.epx

Post-treatment

ECHO

RESU ALIC ’bm_mpi_str_rigi_deco.ali’ GARD PSCR

SORT VISU NSTO 1

FIN

